struct CameraUniform {
    view_pos: vec4<f32>,
    view: mat4x4<f32>,
    view_proj: mat4x4<f32>,
    inv_proj: mat4x4<f32>,
    inv_view: mat4x4<f32>,
}
@group(1) @binding(0)
var<uniform> camera: CameraUniform;

struct LightUniform {
    position: vec3f,
    color: vec3f,
}
@group(2) @binding(0)
var<uniform> light: LightUniform;

// 
struct VertexInput {
    @location(0) position: vec3<f32>,
    @location(1) tex_coords: vec2<f32>,
    @location(2) normal: vec3<f32>,
    @location(3) tangent: vec3<f32>,
    @location(4) bitangent: vec3<f32>,
}

struct InstanceInput {
    @location(5) model_matrix_0: vec4<f32>,
    @location(6) model_matrix_1: vec4<f32>,
    @location(7) model_matrix_2: vec4<f32>,
    @location(8) model_matrix_3: vec4<f32>,
    @location(9) normal_matrix_0: vec3<f32>,
    @location(10) normal_matrix_1: vec3<f32>,
    @location(11) normal_matrix_2: vec3<f32>,
};

struct VertexOutput {
    @builtin(position) clip_position: vec4<f32>,
    @location(0) tex_coords: vec2<f32>,
    @location(1) tangent_position: vec3<f32>,
    @location(2) tangent_light_position: vec3<f32>,
    @location(3) tangent_view_position: vec3<f32>,
}

// 
@vertex
fn vs_main(
    model: VertexInput,
    instance: InstanceInput,
) -> VertexOutput {
    let model_matrix = mat4x4<f32>(
        instance.model_matrix_0,
        instance.model_matrix_1,
        instance.model_matrix_2,
        instance.model_matrix_3,
    );
    let normal_matrix = mat3x3<f32>(
        instance.normal_matrix_0,
        instance.normal_matrix_1,
        instance.normal_matrix_2,
    );
    let world_normal = normalize(normal_matrix * model.normal);
    let world_tangent = normalize(normal_matrix * model.tangent);
    let world_bitangent = normalize(normal_matrix * model.bitangent);
    let tangent_matrix = transpose(mat3x3(
        world_tangent,
        world_bitangent,
        world_normal,
    ));

    let world_position = model_matrix * vec4(model.position, 1.0);

    var out: VertexOutput;
    out.clip_position = camera.view_proj * world_position;
    out.tex_coords = model.tex_coords;
    out.tangent_position = tangent_matrix * world_position.xyz;
    out.tangent_view_position = tangent_matrix * camera.view_pos.xyz;
    out.tangent_light_position = tangent_matrix * light.position;
    return out;
}

// 
@group(0) @binding(0)
var tex: texture_2d<f32>;
@group(0) @binding(1)
var sam: sampler;
@group(0)@binding(2)
var t_normal: texture_2d<f32>;
@group(0) @binding(3)
var s_normal: sampler;

@fragment
fn fs_main(in: VertexOutput) -> @location(0) vec4<f32> {
    let object_normal = textureSample(t_normal, s_normal, in.tex_coords);
    let tangent_normal = object_normal.xyz * 2.0 - 1.0;

    let ambient_strength = 0.1;
    let ambient_color = light.color * ambient_strength;

    let light_dir = normalize(in.tangent_light_position - in.tangent_position);
    // let diffuse_strength = max(dot(in.world_normal, light_dir), 0.0);
    let diffuse_strength = abs(dot(tangent_normal, light_dir));
    let diffuse_color = light.color * diffuse_strength;

    let view_dir = normalize(in.tangent_view_position - in.tangent_position);
    // let reflect_dir = reflect(-light_dir, in.world_normal);
    let half_dir = normalize(view_dir + light_dir);
    // let specular_strength = pow(max(dot(view_dir, reflect_dir), 0.0), 32.0);
    let specular_strength = pow(max(dot(tangent_normal, half_dir), 0.0), 32.0);
    let specular_color = specular_strength * light.color;

    let base_color = vec3<f32>(0.2, 0.2, 0.2);
    let tex_color = textureSample(tex, sam, in.tex_coords);
    let obj_color = mix(base_color, tex_color.rgb, 0.3);

    let result = (ambient_color + diffuse_color + specular_color) * obj_color;
    return vec4<f32>(result, 1.0);
}

// compute shader
/** 
struct InfoParams {
  img_size: vec2<i32>,
};

@group(0) @binding(0) var<uniform> params: InfoParams;
@group(0) @binding(1) var src_pic: texture_2d<f32>;
@group(0) @binding(2) var swap_pic: texture_storage_2d<rgba32float, write>;

const WEIGHT: array<f32, 5> = array<f32, 5>(0.2, 0.1, 0.10, 0.1, 0.1);

@compute @workgroup_size(16, 16)
fn cs_main(@builtin(global_invocation_id) global_id: vec3<u32>) {
  let uv = vec2<i32>(global_id.xy);
  if (uv.x >= params.img_size.x || uv.y >= params.img_size.y) {
    return;
  }

  var temp = textureLoad(src_pic, uv, 0) * WEIGHT[0u];
  let uvMax: vec2<i32> = params.img_size - 1;
  for (var i: i32 = 1; i <= 4; i += 1) {
    var uvOffset = vec2<i32>(3, 0) * i;
    temp += textureLoad(src_pic, clamp(uv + uvOffset, vec2<i32>(0), uvMax), 0) * WEIGHT[i];
    temp += textureLoad(src_pic, clamp(uv - uvOffset, vec2<i32>(0), uvMax), 0) * WEIGHT[i];
  }
  textureStore(swap_pic, uv, temp);
}

*/